Definitions
In mathematics, additive primes are prime numbers for which the sum of their decimal digits are also primes.



Task
Write a program to determine (and show here) all additive primes less than 500.

Optionally, show the number of additive primes.



Also see

   the OEIS entry:   A046704 additive primes.
   the prime-numbers entry:   additive primes.
   the geeks for geeks entry: additive prime number.
   the prime-numbers fandom: additive primes.


10 M=499
20 :1)=1
30 P=2
40 :P)=0
50 P=P+1
60 #=M>P*40
70 P=2
80 C=P*2
90 :C)=1
110 C=C+P
120 #=M>C*90
130 P=P+1
140 #=M/2>P*80
150 P=2
160 N=0
170 #=:P)*290
180 S=0
190 K=P
200 K=K/10
210 S=S+%
220 #=0<K*200
230 #=:S)*290
240 ?=P
250 $=9
260 N=N+1
270 #=N/10*0+%=0=0*290
280 ?=""
290 P=P+1
300 #=M>P*170
310 ?=""
320 ?="There are ";
330 ?=N
340 ?=" additive primes below ";
350 ?=M+1